Series Solutions

Series solutions near an ordinary point for a second order, linear, nonhomogenous DE $(1-x)\, y'' + y = 3\, x^2 - x + 4$

■ what can we tell from the DE about a solution?

Note that in standard form, the DE is y'' + (0) y' + $\left(\frac{1}{1-x}\right)$y = $\left(\frac{3\,x^2-x+4}{1-x}\right)$. The coefficients are p(x) = 0, q(x) = $\frac{1}{1-x}$, g(x) =$\frac{3\,x^2-x+4}{1-x}$. These are analytic (i.e.,Taylor Series with a positive radius of convergence, i.e., have an infinite number of continuous derivatives) at every point of the real line except x_0=1. In other words, x_0=1 is a singular point of the DE and every other point of the real line is an ordinary point. In particular, x_0=0 is an ordinary point, and we may seek a solution of the form y(x) = $\sum_{n=0}^{\infty} a_n\, x^n$. Theory only guarantees that such a series solution converges on the interval (-1,1). The solution may be good outside of this interval.

■ 3 terms in each of 2 linearly independent solns

First state the form of the solution to a desired number of terms.

```
Clear[y, a, y1, y2, i, x, lhs, ysoln]
```

```
y=Sum[a[i] x^i, {i,0,6}] + O[x]^7
```

$a[0] + a[1]\, x + a[2]\, x^2 + a[3]\, x^3 + a[4]\, x^4 + a[5]\, x^5 + a[6]\, x^6 + O[x]^7$

Calculate the necessary derivatives

```
y1=D[y,x]
```

$a[1] + 2\, a[2]\, x + 3\, a[3]\, x^2 + 4\, a[4]\, x^3 + 5\, a[5]\, x^4 + 6\, a[6]\, x^5 + O[x]^6$

```
y2=D[y1,x]
```

$2\, a[2] + 6\, a[3]\, x + 12\, a[4]\, x^2 + 20\, a[5]\, x^3 + 30\, a[6]\, x^4 + O[x]^5$

Substitute y and its derivatives into the differential equation.

```
lhs=Expand[(1-x)*y2+y]
```

$(a[0] + 2\, a[2]) + (a[1] - 2\, a[2] + 6\, a[3])\, x + (a[2] - 6\, a[3] + 12\, a[4])\, x^2 + (a[3] - 12\, a[4] + 20\, a[5])\, x^3 + (a[4] - 20\, a[5] + 30\, a[6])\, x^4 + O[x]^5$

Equate the coefficients for corresponding powers of x.

```
LogicalExpand[lhs == 3 x^2 - x + 4]
```

$-4 + a[0] + 2\, a[2] == 0\ \&\&\ 1 + a[1] - 2\, a[2] + 6\, a[3] == 0\ \&\&\ -3 + a[2] - 6\, a[3] + 12\, a[4] == 0\ \&\&\ a[3] - 12\, a[4] + 20\, a[5] == 0\ \&\&\ a[4] - 20\, a[5] + 30\, a[6] == 0$

Since the DE is second order, we should be able to solve for all but two terms in the series solution. Since the DE is nonhomogeneous, we are not guaranteed to find all in terms of the first two, but it works here.

```
Solve[%,{a[2],a[3],a[4],a[5],a[6]}]
```

$$\left\{\left\{a[2]\to\frac{1}{2}(4-a[0]),\ a[3]\to\frac{1}{6}(3-a[0]-a[1]),\ a[4]\to\frac{1}{24}(8-a[0]-2\,a[1]),\ a[5]\to\frac{1}{120}(21-2\,a[0]-5\,a[1]),\ a[6]\to\frac{1}{720}(76-7\,a[0]-18\,a[1])\right\}\right\}$$

Substitute these back into y.

```
{ysoln}=y/.%
```

$$\left\{a[0]+a[1]\,x+\frac{1}{2}(4-a[0])\,x^2+\frac{1}{6}(3-a[0]-a[1])\,x^3+\frac{1}{24}(8-a[0]-2\,a[1])\,x^4+\frac{1}{120}(21-2\,a[0]-5\,a[1])\,x^5+\frac{1}{720}(76-7\,a[0]-18\,a[1])\,x^6+O[x]^7\right\}$$

Collect the terms for the two linearly independent series solutions; we're only showing a certain number of terms in each.

```
ysoln=Collect[ysoln,{1, a[0],a[1]}]
```

$$2\,x^2+\frac{x^3}{2}+\frac{x^4}{3}+\frac{7\,x^5}{40}+\frac{19\,x^6}{180}+\left(1-\frac{x^2}{2}-\frac{x^3}{6}-\frac{x^4}{24}-\frac{x^5}{60}-\frac{7\,x^6}{720}\right)a[0]+\left(x-\frac{x^3}{6}-\frac{x^4}{12}-\frac{x^5}{24}-\frac{x^6}{40}\right)a[1]$$

Look at the portions of this solution. The linearly independent solutions to the associated homogenous equation are $y1(x) = 1-\frac{x^2}{2}-\frac{x^3}{6}-\frac{x^4}{24}-\frac{x^5}{60}-\frac{7x^6}{720}+...$ and $y2(x) = x-\frac{x^3}{6}-\frac{x^4}{12}-\frac{x^5}{24}-\frac{x^6}{40}+...;$

the particular solution is $yp(x) = 2\,x^2+\frac{x^3}{2}+\frac{x^4}{3}+\frac{7x^5}{40}+\frac{19x^6}{180}+...$

```
In[30]:= u1 = Coefficient[ysoln, a[0]];
u2 = Coefficient[ysoln, a[1]];
yp = ysoln - u1 a[0] - u2 a[1];
```

■ some of these partial sums

```
ysoln /. {x → 0}
```

a[0]

What is y' (0)?

```
D[ysoln, x]
```

$$4\,x+\frac{3\,x^2}{2}+\frac{4\,x^3}{3}+\frac{7\,x^4}{8}+\frac{19\,x^5}{30}+\left(-x-\frac{x^2}{2}-\frac{x^3}{6}-\frac{x^4}{12}-\frac{7\,x^5}{120}\right)a[0]+\left(1-\frac{x^2}{2}-\frac{x^3}{3}-\frac{5\,x^4}{24}-\frac{3\,x^5}{20}\right)a[1]$$

In this case, a[0] is y(0), the y-intercept, and a[1] is the slope of the curve at that point.

```
D[ysoln, x] /. {x → 0}
```

a[1]

Do not worry about the code here; we generate some initial values and some solution curves. Here we have chosen some points with different y-intercepts and nonnegative slopes at the initial value of x=0.

In[24]:=
```
Clear[pts, ptList, fcnList, c1, c2];
ptList = {}; fcnList = {};

Do[AppendTo[ptList, {0, c1}], {c1, -4, 4, 2}]
For[c1 = -4, c1 ≤ 4,
 For[c2 = 0, c2 ≤ 3, c2++, AppendTo[fcnList, u1 c1 + u2 c2 + yp] ];
 c1 = c1 + 2]
pts = Graphics[{PointSize[Large], Point[ptList]}];
Show[{Plot[fcnList, {x, -3, 4}, PlotRange → {-9, 11}], pts}]
```

Out[29]=



■ Aiding in handwork

y(x) = $\sum_{n=0}^{\infty} a_n\, x^n$. y'(x) = $\sum_{n=1}^{\infty} n\, a_n\, x^{n-1}$. y''(x) = $\sum_{n=2}^{\infty} n(n-1)\, a_n\, x^{n-2}$.

3 x^2 - x + 4 = (1 - x) y''(x) + y(x) = $\sum_{n=2}^{\infty} n(n-1)\, a_n\, x^{n-2}$ - x $\sum_{n=2}^{\infty} n(n-1)\, a_n\, x^{n-2}$ + $\sum_{n=0}^{\infty} a_n\, x^n$

= $\sum_{n=2}^{\infty} n(n-1)\, a_n\, x^{n-2}$ - $\sum_{n=2}^{\infty} n(n-1)\, a_n\, x^{n-1}$ + $\sum_{n=0}^{\infty} a_n\, x^n$ (let j=n-2 in series 1, j=n-1 in series 2, j=n in series 3)

= $\sum_{j=0}^{\infty} (j+2)\,(j+1)\, a_{j+2}\, x^j$ - $\sum_{j=-0}^{\infty} (j+1)\, j\, a_{j+1}\, x^j$ + $\sum_{j=0}^{\infty} a_j\, x^j$

= $\sum_{j=0}^{\infty} \left\{ (j+2)\,(j+1)\, a_{j+2} - (j+1)\, j\, a_{j+1} + a_j \right\} x^j$

coefficient of x^0

```
Solve[4 == 2 a[2] + a[0], a[2]]
```

$\left\{\left\{a[2] \to \frac{1}{2}\,(4 - a[0])\right\}\right\}$

```
Expand[%]
```

$\left\{\left\{a[2] \to 2 - \frac{a[0]}{2}\right\}\right\}$

```
subs1 = {a[2] → 2 - a[0]/2}
```

$\left\{a[2] \to 2 - \frac{a[0]}{2}\right\}$

```
(a[2 + j] (1 + j) (2 + j) - j (1 + j) a[1 + j] + a[j]) /. {j → 1}
```

$a[1] - 2\, a[2] + 6\, a[3]$

coefficient of x^1

```
Solve[-1 == a[1] - 2 a[2] + 6 a[3], a[3]]
```

$\left\{\left\{a[3] \to \frac{1}{6} (-1 - a[1] + 2\, a[2])\right\}\right\}$

```
Expand[%]
```

$\left\{\left\{a[3] \to -\frac{1}{6} - \frac{a[1]}{6} + \frac{a[2]}{3}\right\}\right\}$

```
% /. subs1
```

$\left\{\left\{a[3] \to -\frac{1}{6} + \frac{1}{3}\left(2 - \frac{a[0]}{2}\right) - \frac{a[1]}{6}\right\}\right\}$

```
Expand[%]
```

$\left\{\left\{a[3] \to \frac{1}{2} - \frac{a[0]}{6} - \frac{a[1]}{6}\right\}\right\}$

$\mathbf{subs2} = \left\{a[2] \to 2 - \frac{a[0]}{2},\ a[3] \to \frac{1}{2} - \frac{a[0]}{6} - \frac{a[1]}{6}\right\}$

$\left\{a[2] \to 2 - \frac{a[0]}{2},\ a[3] \to \frac{1}{2} - \frac{a[0]}{6} - \frac{a[1]}{6}\right\}$

```
(a[2 + j] (1 + j) (2 + j) - j (1 + j) a[1 + j] + a[j]) /. {j → 2}
```

$a[2] - 6\, a[3] + 12\, a[4]$

coefficient of x^2

```
Solve[3 == a[2] - 6 a[3] + 12 a[4], a[4]]
```

$\left\{\left\{a[4] \to \frac{1}{12} (3 - a[2] + 6\, a[3])\right\}\right\}$

```
% /. subs2
```

$\left\{\left\{a[4] \to \frac{1}{12}\left(1 + \frac{a[0]}{2} + 6\left(\frac{1}{2} - \frac{a[0]}{6} - \frac{a[1]}{6}\right)\right)\right\}\right\}$

```
Expand[%]
```

$\left\{\left\{a[4] \to \frac{1}{3} - \frac{a[0]}{24} - \frac{a[1]}{12}\right\}\right\}$

$\mathbf{subs3} = \left\{a[2] \to 2 - \frac{a[0]}{2},\ a[3] \to \frac{1}{2} - \frac{a[0]}{6} - \frac{a[1]}{6},\ a[4] \to \frac{1}{3} - \frac{a[0]}{24} - \frac{a[1]}{12}\right\}$

$\left\{a[2] \to 2 - \frac{a[0]}{2},\ a[3] \to \frac{1}{2} - \frac{a[0]}{6} - \frac{a[1]}{6},\ a[4] \to \frac{1}{3} - \frac{a[0]}{24} - \frac{a[1]}{12}\right\}$

coefficient of x^n for n>=3

Solve[a[2 + j] (1 + j) (2 + j) - j (1 + j) a[1 + j] + a[j] == 0, a[2 + j]]

$$\left\{\left\{a[2+j]\to \frac{-a[j]+j\,a[1+j]+j^2\,a[1+j]}{2+3\,j+j^2}\right\}\right\}$$

Simplify[%]

$$\left\{\left\{a[2+j]\to \frac{-a[j]+j\,(1+j)\,a[1+j]}{2+3\,j+j^2}\right\}\right\}$$

Factor[%]

$$\left\{\left\{a[2+j]\to \frac{-a[j]+j\,a[1+j]+j^2\,a[1+j]}{(1+j)\,(2+j)}\right\}\right\}$$

$$\left\{a[2+j]\to \frac{-a[j]+j\,(j+1)\,a[1+j]}{(1+j)\,(2+j)}\right\}\;/.\;\{j\to 3\}$$

$$\left\{a[5]\to \frac{1}{20}\,(-a[3]+12\,a[4])\right\}$$

Expand[%]

$$\left\{a[5]\to -\frac{a[3]}{20}+\frac{3\,a[4]}{5}\right\}$$

% /. subs3

$$\left\{a[5]\to \frac{3}{5}\left(\frac{1}{3}-\frac{a[0]}{24}-\frac{a[1]}{12}\right)+\frac{1}{20}\left(-\frac{1}{2}+\frac{a[0]}{6}+\frac{a[1]}{6}\right)\right\}$$

Expand[%]

$$\left\{a[5]\to \frac{7}{40}-\frac{a[0]}{60}-\frac{a[1]}{24}\right\}$$

$$\text{subs4} = \left\{a[2]\to 2-\frac{a[0]}{2},\; a[3]\to \frac{1}{2}-\frac{a[0]}{6}-\frac{a[1]}{6},\; a[4]\to \frac{1}{3}-\frac{a[0]}{24}-\frac{a[1]}{12},\; a[5]\to \frac{7}{40}-\frac{a[0]}{60}-\frac{a[1]}{24}\right\}$$

$$\left\{a[2]\to 2-\frac{a[0]}{2},\; a[3]\to \frac{1}{2}-\frac{a[0]}{6}-\frac{a[1]}{6},\; a[4]\to \frac{1}{3}-\frac{a[0]}{24}-\frac{a[1]}{12},\; a[5]\to \frac{7}{40}-\frac{a[0]}{60}-\frac{a[1]}{24}\right\}$$

$$\left\{a[2+j]\to \frac{-a[j]+j\,(j+1)\,a[1+j]}{(1+j)\,(2+j)}\right\}\;/.\;\{j\to 4\}$$

$$\left\{a[6]\to \frac{1}{30}\,(-a[4]+20\,a[5])\right\}$$

Expand[%]

$$\left\{a[6]\to -\frac{a[4]}{30}+\frac{2\,a[5]}{3}\right\}$$

% /. subs4

$$\left\{a[6]\to \frac{2}{3}\left(\frac{7}{40}-\frac{a[0]}{60}-\frac{a[1]}{24}\right)+\frac{1}{30}\left(-\frac{1}{3}+\frac{a[0]}{24}+\frac{a[1]}{12}\right)\right\}$$

```
Expand[%]
```

$$\left\{a[6] \to \frac{19}{180} - \frac{7\,a[0]}{720} - \frac{a[1]}{40}\right\}$$